

ARRL Handbook CD

Template File

Title: Azimuth and Elevation Information

Chapter: 23

Topic: EME

Template contains:

Azimuth and elevation information for half-hour intervals during a UTC day.

Table 10

Computer Program for Determining Azimuth and Elevation of Celestial Bodies

```
10 REM * * * MOONTRAK.BAS * * *
20 CLS:PRINT"PROGRAM TO CALCULATE AZ-EL DATA FOR THE SUN OR MOON"
30 PRINT:PRINT"Program by J. Hall, K1TD, ARRL Hq., Rev 1.1, June 1983":PRINT
40 PRINT"This program may be reproduced without prior permission"
50 PRINT"provided The ARRL Handbook is credited.":PRINT
60 B$="###.#":I=57.2958:K=.01657
70 PRINT"Enter negative values for southerly latitudes."
80 PRINT"Enter negative values for easterly longitudes."
90 PRINT:INPUT"Your latitude (degrees and decimal)";A
100 INPUT"Your longitude (degrees and decimal)";L1
110 D=SIN(A/I):F=COS(A/I)
120 INPUT"UTC date (no comma, please)";A$
130 INPUT"Data for which, sun or moon";C$:PRINT
140 PRINT A$ " GHA of ";C$;" at 0000 UTC":INPUT" (degrees and decimal)";L2
150 PRINT A$ " declination at 0000 UTC":INPUT" (degrees and decimal)";B1
160 PRINT A$ " GHA at 2400 UTC":INPUT" (degrees and decimal)";L3
170 PRINT A$ " declination at 2400 UTC":INPUT" (degrees and decimal)";B2
180 GI=(L3+360-L2)/24
190 BI=(B2-B1)/24
200 J%=0:GOSUB 330
210 FOR A%=0 TO 48:G=SIN(B1/I):L=L2-L1
220 E=D*G+F*COS(B1/I)*COS(L/I):C=(G-D*E)/F:J=E
230 IF E>=1 THEN E=1.5708:GOTO 250
240 IF E<=-1 THEN E=-1.5708 ELSE E=ATN(E/SQR(ABS(1-E*E)))
250 C=C/COS(E):IF C>=1 THEN C=0:GOTO 270
260 IF C<=-1 THEN C=180 ELSE C=I*(-ATN(C/SQR(ABS(1-C*C))))+1.5708
270 IF LEFT$(C$,1)="m" OR LEFT$(C$,1)="M" THEN GOSUB 430
280 E=I*E:IF J%=8 AND E>=0 THEN GOSUB 410
290 IF E>=0 THEN GOSUB 360
300 B1=B1+BI/2:L2=L2+GI/2:IF L2>360 THEN L2=L2-360
310 NEXT A:PRINT:PRINT"Data for ";A$;" is completed for the ";C$
320 INPUT"To continue, press enter";J$:CLS:PRINT"Next ";GOTO 120
330 CLS:PRINT"Data for the ";C$" from "A;"deg. lat., ";L1;"deg. long."
340 PRINT"for ";A$:PRINT
350 PRINT"Time, UTC", "Azimuth", "Elevation":PRINT:RETURN
360 IF INT(A%/2)=A%/2 THEN D$=STR$(50*A%) ELSE D$=STR$(50*A%-20)
370 D$=RIGHT$(D$,LEN(D$)-1)
380 IF LEN(D$)<4 THEN D$="0"+D$:GOTO 380
390 PRINT D$,":IF SIN(L/I)<=0 THEN PRINT USING B$;C; ELSE PRINT USING B$;360-C;
400 PRINT TAB(32);:PRINT USING B$;E:J%=J%+1:RETURN
410 PRINT:INPUT"For more data press enter";J%
420 J%=0:GOSUB 330:RETURN
430 J=(J-K)/COS(E):E=ATN(J):RETURN
440 REM NOTE: % = integer
```